
Figure 2   VM-level container meta-model 

distribution transparency. As a result of this 
provision, the ASSERT designer should not need to 
modify source code to deploy different 
configurations of the system. 

The communication services that interact with the physical 
interconnection might also impose some requirements to be 
fulfilled by the middleware. The analysis of its full impact 
is currently in progress, due to complete by Q3 2007. As an 
adaptation layer, the middleware shall limit its weight to 
avoid any unjustified overhead, while preserving structural 
properties and proof preservation. 

3   Code Patterns for local VM-Level 
Containers 
Local VMLC are run-time entities that provide direct 
support for concurrent real-time behaviour based on the 
Ravenscar computation model. They are directly imple-
mented by primitive services of the ASSERT VM, using 
the above described extension of the Ada Ravenscar 
profile. Four basic types of VMLC have been defined, 
based on the hard real-time terminal objects of HRT-
HOOD [1]: cyclic, sporadic, protected, and passive 
containers. 

VMLC are implemented using some meta-model elements, 
which are named after HRT-HOOD entities (figure 2): 

• Provided and required interfaces, 

• OPCS (operation control structure), embodying  the 
functional behaviour of the component; 

• OBCS (object control structure), providing 
synchronization mechanisms; 

• Thread, providing the concurrent behaviour of the 
component. 

The real-time and synchronization behaviour of a VMLC is 
provided by the OBCS and the thread. The OBCS is im-
plemented by an Ada protected object, while the thread is 
implemented by an Ada task. 

Code patterns that represent the common aspects of a class 
of model elements are called archetypes. Examples of 
archetypes are: 

• Cyclic object, with or without deadline and overrun 
detection; 

• Sporadic object, with or without deadline and 
overrun detection; 

Appendix A includes some examples of such archetypes. 
More complex archetypes can built by combining these 
code patterns, e.g. a cyclic thread with both deadline miss 
and WCET overrun detection. 

A key issue is how to handle temporal faults. Possible fault 
handling patterns include: 

• Error logging. Although useful for testing or low 
integrity applications, this pattern is not suited for 
high-integrity tasks as it does not provide any kind 
of resilience to faults. 

• Second chance. If there is some slack time available, 
the execution-time budget of the failing task can be 
extended for the current job. This is seldom 
acceptable in high-integrity systems, as it incurs 
some degree of temporal indeterminacy. 

• Mode change. This is the most flexible approach in 
that it permits fault recovery to be performed as and 
when required. 

• Safe stop. This strategy is often the only course to 
take for high-integrity systems. 

A more extensive discussion of these schemes can be found 
in an upcoming paper [8]. 

4   Extending Patterns for Distribution 
In this section we review Ada coding patterns for VMLC in 
the distributed case. 

The support of distribution encompasses a variety issues 
that range from high-level middleware architecture down to 
implementation concerns. Some of those issues are al-ready 
discussed from an Ada perspective in PolyORB [13]; GLADE 
[7] and RT-GLADE projects. For the purposes of this 
paper we focus on interaction patterns, which are the most 
critical elements to warrant system analyzability. 

We make the following hypotheses: a high-level model 
describes the deployment of the system; this view is used to 
build and dimension all tasks and buffers, in-place mar-
shallers are built; a deterministic protocol is used 1. Hence, we 
                                                           
1 Please refer to the authors bibliography for more details on these 
hypotheses. 

focus on how interaction patterns defined in the Ravenscar 
profile can be mapped onto interaction suitable for 
distribution. 

We assume interactions are asynchronous and one-way, 
which is a sensible choice for Distributed Real-Time systems for 
it reduces global blocking time. Besides, this is compatible with 
typical requirements from critical systems and typical scheduling 
techniques such as [12] and its extensions. We assume that a one-
to-one relation exists between the priority of the request (the 
message to be exchanged) and the priority of the processing 
tasks. 

The Ravenscar profile prescribes that inter-task communication 
take place via protected objects. Extending this configuration to 
distribution can be as simple as “splitting” the protected object in 
two separate entities: one client-side activity that formats the 
request and passes it to the communication subsystem; one server-
side activity that is notified a request is arriving. Depending on the 
enforcement policies of choice, a task can be released from a 
thread pool, or else a dedicated task is awakened. Let us briefly 
review each configuration in isolation (figure 3): 

• Client Side: this configuration is similar to the local 
case: the client passes its request to the 
communication subsystem that sends this request to 
the remote node. Intermediate steps are required to 
marshall the payload, and to build proper messages. 
Depending on the protocol stack, an intermediate 
protected object might be placed between client 
code and the protocol-specific threads. 

• Server Side: on the server side, the request is 
received by the communication subsystem and 
stored in some internal buffer. It is safe to assume 
that this buffer is embodied in a protected object 
(PO). Several scenarios are then possible: 

- One PO per receiving task (1): this scenario is 
similar to the local case where at most one task 
is waiting per protected object; 

- One PO plus a pool of tasks (2): this scenario 
implies that a task is waiting for the next 
request while some other tasks are either 
processing jobs or idling, or else blocked on a 
task-specific PO. This arrangement corres-
ponds to the Leader/Followers or Thread Pool 
patterns [10]. In such a scenario it may be 
convenient to use different priorities for I/O 
and message sending as well as for request 
processing tasks, for example to discriminate 
between urgency and priority. 

Furthermore, one must select patterns to dispatch to the 
correct subprogram, with two strategies one PO per sub-
program, or one PO shared by several subprograms and 
skeleton-like code to dispatch to the correct subprogram. 
Each strategy has been implemented and made deterministic 
separately by using local Ravenscar patterns and a transport 
library. Those two elements can then be combined to form four 
different configurations. The selection of one particular  
configuration has to be made depending on two antonymous  

 

 

Figure 3   Request Handling Strategies 

criteria: footprint size of the system vs. ease of analysis. At the 
time of this writing the ASSERT project team is still evaluating 
the costs and trade-offs of each configuration in term of 
pessimistic WCET and memory overhead. 

5   Conclusions and future work 
In this paper we have briefly outlined the motivations and 
the strategy we have adopted for the definition of Ada 2005 
coding patterns for the support of the code generation stage 
of an advanced model-driven development infrastructure. 
We are presently approaching the final implementation 
stage of our vision and we shall soon have performance 
figures useful to ascertain the viability of the approach we 
pursue. 

A   VM-level container archetypes 
The code templates in the next pages illustrate the approach 
use in building component archetypes. More complex 
archetypes cab be built using this basic set. 

Pattern 1. Cyclic thread archetype with deadline miss detection. 

 Deadline_Miss : Timing_Event ; 
 task type  Cyclic_Thread(P :  Priority; T : Time_Span ; D : Time_Span )  is 
  pragma  Priority(P) ; 
 end  Cyclic_Thread; 
 task body  Cyclic_Thread  is 
  Next_Time : Time : = Clock ; 
  Deadline_Missed : Boolean ; 
 begin 
  loop 
   Deadline_Miss.Set_Handler ( Next _Time + D, Deadline_Miss _Handler ’ Access ) ; 
   delay until  Next _Time ; 
   OPCS. Activity ;  −−  periodic job 
   Next_Time : = Next_Time + T ; 
  end loop ; 
 end  Cyclic_Thread; 
 
 
 

Pattern 2. Cyclic thread archetype with WCET overrun detection. 
 task type  Cyclic_Thread(P :  Priority; T : Time_Span ; D : Time_Span )  is 
  pragma  Priority(P) ; 
 end  Cyclic_Thread; 
 task body  Cyclic_Thread  is 
  Next_Time : Time : = Clock ; 
  Id :  aliased constant  Task_Id : = Current_Task ; 
  WCET_Timer : Ada.Execution_Time.Timers .Timer ( Id ’ Access ) ; 
 begin 
  loop 
   Set_Handler ( WCET_Timer , C, Overrun_Handler ’ Access ) ; 
   delay until Next_Time ; 
   OPCS. Activity ;  −−  periodic job 
   Next_Time : = Next_Time + T ; 
  end loop ; 
 end  Cyclic_Thread; 
 
 
 

Pattern 3. Sporadic thread archetype with minimal inter-arrival time enforcement. 
 OBCS : Sporadic_OBCS ( C e i l i n g ) ; 
 task type  Sporadic_Thread(P :  Priority; T : Time_Span)  is 
  pragma  Priority(P) ; 
 end  Sporadic_Thread; 
 task body  Sporadic_Thread  is 
  Next_Time : Time : = Clock ; 
 begin 
  loop 
   delay until Next_Time ; 
   OBCS. Get_Request ; 
   OPCS. Activity ;  −−  sporadic job   
   Next_Time : = Next_Time + T ; 
  end loop ; 
 end  Cyclic_Thread; 
 procedure  Start  is 
 begin 
  OBCS.Put_Request ; 
 end  Start ; 

Ada Gems 
The following contributions are taken from the AdaCore Gem of the Week series. The full collection of gems, discussion and 
related files, can be found at http://www.adacore.com/category/developers-center/gems/. 

 

Ada Gem -5: Key-Based Searching 
In Set Containers 
Matthew Heaney, On2 Technologies 
Date: 11 June 2007 
 
Abstract: Sets are containers of elements. Like all containers, 
they are searchable, and given an element value you can find 
the element in the set equivalent to that value. For some 
applications, element equivalence is determined by just a part 
of the element (its “key” part), and it is often necessary to find 
an element given only a key value. A technique is presented 
here for key-based searching of elements in set containers.  
 
Let’s get started… 
This example demonstrates how to use hashed set containers.  
We are given the task of implementing a game in which a user 
navigates among components in a maze. These map-sites 
comprise walls, rooms, and doors. One of the features of the 
game is the ability to find room objects given a room number, 
and we implement that feature using a set container. 
The map-site objects in our simulation are members of a 
common class. We use a tagged type declared as follows: 

   type Map_Site is abstract tagged limited null record; 

The (abstract) type has a single (abstract) operation to provide 
navigation to a site object:  

   procedure Enter (Site : in out Map_Site) is abstract; 

Next we implement concrete type Room, which derives from 
Map_Site. The declaration of the Room type looks like this:  

   type Room (<>) is limited new Map_Site with private; 

The room type is limited and indefinite (because it has 
“unknown discriminants”), which means that objects of the 
type must be explicitly initialized by calling a constructor-style 
function, declared as follows: 

   not overriding 
   function New_Room (Number : Positive) return not null  
               access Room; 

There is no way to create a room object other than to call 
New_Room, which is where we insert new room objects into 
the set used to find instances. Now we turn to the package 
body and the instantiation of the set package. The generic 
actual type is just a simple named access type (the same one 
we use to allocate instances): 

   type Room_Access is not null access Room; 

To instantiate the (hashed) set package, we need both a hash 
function and an equivalence function. The hash function for 
sets is a mapping of element value to hash value. So how do 
we make a hash value from a room object? The natural choice 

is to use the room number as the hash value for a room object, 
so our hash function for rooms looks like this: 

   function Hash (R : Room_Access) return Hash_Type is 
   begin 
       return Hash_Type (R.Number); 
   end; 

For the equivalence function, we can just use the predefined 
equality operator for type Room_Access, since set elements 
are access values, and access values are unique for distinct 
room objects. 
We now have everything we need to instantiate the generic 
hashed set container: 

   package Room_Sets is new  
              Ada.Containers.Hashed_Sets 
              (Element_Type             => Room_Access, 
               Hash                            => Hash, 
               Equivalent_Elements  => “=”); 
 
   Room_Set : Room_Sets.Set; 

We are finally in a position to implement the Room 
constructor function, which looks like this: 

   function New_Room (Number : Natural)  
      return not null access Room  
   is 
      R : constant Room_Access := new Room (Number); 
   begin 
      Room_Set.Insert (R); 
      return R; 
   end New_Room; 

Now that we have implemented the necessary infrastructure 
for creating room objects, we have to implement the function 
for looking up room objects using the room number as the key. 
Our function is declared as follows: 

   function Find_Room (Number : Natural) return  
              access Room; 

Now comes the interesting part. The set records all Room 
objects that have been created, so if a room with that number 
exists, it will be in the set. Our problem is that we have no 
immediate way to search for room objects given a room 
number. Remember that this is a set of rooms (not a map with 
room number as the key) and the search function for the 
instantiation looks like this: 

   function Find (Container : Set; Item : Room_Access)  
              return Cursor; 

The issue is that we have a room number, not a room object, 
so how do we look up a room object according to its number? 
The solution is to take advantange of the nested generic 
package Generic_Keys provided by the sets, which allows 
you to view a set element in terms of its key. It has generic 
formals very similar to the set package itself, except that the 

operations apply to a generic formal Key_Type instead of to 
the element type. 
One requirement for using that package is that the generic 
actuals for keys must deliver the same values as the generic 
actuals for elements. For example, the function to return the 
hash value of a room number (the key) must return the same 
value as the function that returns the hash value of the room 
object (the element) with that room number. So we instantiate 
the nested package as follows: 

   function Get_Room_Number (R : Room_Access)  
              return Natural is 
   begin 
      return R.Number; 
   end; 
   function Room_Number_Hash (N : Natural)  
              return Hash_Type is 
   begin 
      return Hash_Type (N); 
   end; 
 
   package Room_Number_Keys is new  
              Room_Sets.Generic_Keys  
              (Key_Type            => Natural,  
               Key                      => Get_Room_Number,  
               Hash                    => Room_Number_Hash,  
               Equivalent_Keys => “=”); 

We are now ready to implement the Find_Room function. The 
package Room_Number_Keys provides a key-based search 
function, so we call that to search the set for a room object 
with the given room number: 

   function Find_Room (Number : Natural)  
              return access Room is 
      C : constant Room_Sets.Cursor := 
            Room_Number_Keys.Find (Room_Set, Number); 
   begin 
      if Has_Element (C) then 
         return Element (C); 
      else 
         return null; 
      end if; 
   end Find_Room; 

The search function returns a cursor, and the cursor value 
indicates whether the search was successful. If a room object 
having that room number is in the set, Has_Element returns 
True and so we return the Room object designated by the 
cursor. If that number was not found (because no room object 
having that number was in the set), we simply return null. 
The package Generic_Keys has a few other interesting 
features, including the ability to modify set elements, but we’ll 
defer that discussion until a future time. In the meantime, have 
fun with the containers! 

Ada Gem -8: Factory Functions 
Matthew Heaney, On2 Technologies 
Date: 10 September 2007 
 
Abstract:  A factory function is a technique for constructing 
an object from one class given only an object in some other 

class. The technique is used to implement assignment for 
objects having a class-wide type, such that tag-mismatch 
exceptions cannot occur.  
 
Let’s get started… 
Suppose we have a generic package that declares a stack class. 
The root of the hierarchy would be as follows: 

   generic 
      type Element_Type is private; 
   package Stacks is 
      type Stack is abstract tagged null record; 
 
      procedure Push (Container : in out Stack; 
                                   Item : in Element_Type) is abstract; 
      … 
   end Stacks; 

Assume there are various concrete types in the class, say an 
unbounded stack (that automatically grows as necessary) and a 
bounded stack (implemented as a fixed-size array). 
Now suppose we want to assign one stack to another, 
irrespective of the specific stack type, something like this: 

   procedure Op (T : in out Stack’Class; S : Stack’Class) is 
   begin 
      T := S;  -- raises exception if tags don’t match 
      … 
   end; 

This compiles, but isn’t very robust, since if the tag of the 
target stack doesn’t match the tag of the source stack, then an 
exception will occur. Our goal here is to figure out how to 
assign stack objects (whose type is class-wide) in a manner 
such that the assignment is guaranteed to work without raising 
a tag-mismatch exception. 
One way to do this is to make an assignment-style operation 
that is primitive for the type, so that it will dispatch according 
to the type of the target stack. If the type of the source stack is 
class-wide, then there can’t be a tag mismatch (and hence no 
exception) since there’s only one controlling parameter. 
(Note that you could do it the other way too, by dispatching on 
the tag of the source stack. You could even make the operation 
class-wide, so that it doesn’t need to dispatch at all. The idea is 
to avoid passing more than a single controlled operand.) 
The assign operation would be declared like this: 

   procedure Assign (Target : in out Stack; 
                                 Source : Stack’Class) is abstract; 

which would allow us to rewrite the above assignment 
statement as: 

   procedure Op (T : in out Stack’Class; S : Stack’Class) is 
   begin 
      T.Assign (S);  -- dispatches according T’s tag 
      … 
   end; 

Each type in the class will have to override Assign. As an 
example, let’s follow the steps the necessary to implement the 
operation for the bounded stack type. Its spec would look like 
this: 

   generic 
   package Stacks.Bounded_G is 

      type Stack (Capacity : Natural) is  
           new Stacks.Stack with private; 
 
      procedure Assign 
       (Target : in out Stack; 
        Source : Stacks.Stack’Class); 
      … 
   private 
      type Stack (Capacity : Natural) is 
           new Stacks.Stack with  
      record 
         Elements  : Element_Array (1 .. Capacity); 
         Top_Index : Natural := 0; 
      end record; 
   end Stacks.Bounded_G; 

This is just a canonical implementation of a bounded container 
form, that uses a discriminent to control how much storage for 
the object is allocated. The interesting part is implementing the 
Assign operation, since we need some way to iterate over 
items in the source stack. Here’s a skeleton of the 
implementation: 

   procedure Assign  
                      (Target : in out Stack;  -- bounded form 
                        Source : Stacks.Stack’Class) 
   is 
      … 
   begin 
      … 
      for I in reverse 1 .. Source.Length loop 
         Target.Elements (I) := <get curr elem of source> 
         <move to next elem of source> 
      end loop; 
      …       
   end Assign; 

Note carefully that, assuming we visit items of the source 
stack in top-to-bottom order, it’s not a simple matter of 
pushing items onto the target stack, since if we did that the 
items would end up in reverse order. That’s the reason why we 
populate the target stack array in reverse, starting from largest 
index (the top of the stack) and working backwards (towards 
the bottom of the stack). 
The question is, how do you iterate over the source stack? 
Assume that each specific type in the stack class has its own 
iterator type, matched to that stacks’s particular representation 
(similar to how the containers in the standard library are 
implemented). The issue is that the type of the source stack 
formal parameter is class-wide. How do we get an iterator for 
the source stack actual parameter, if its specific type is not 
known (not known statically, that is)? 
The answer is, just ask the stack for one! A tagged type has 
dispatching operations, some of which can be functions, so 
here we just need a dispatching function to return an iterator 
object. The idiom of dispatching on an object whose type is in 
one class, to return an object whose type is in another class, is 
called a “factory function” or “dispatching constructor.” 
An operation can only be primitive for one tagged type, so if 
the operation dispatches on the stack parameter then the 
function return type must be class-wide. We now introduce 
type Cursor, the root of the stack iterator hierarchy, and amend 
the stack class with a factory function for cursors: 

   type Cursor is abstract tagged null  
        record;  -- the iterator 
 
   function Top_Cursor  -- the factory function 
     (Container : not null access constant Stack) 
     return Cursor’Class is abstract; 
 
   … -- primitive ops for the Cursor class 

Each type in the stack class will override Top_Cursor, to 
return a cursor that can be used to visit the items in that stack 
object. We can now complete our implementation of the 
Assign operation for bounded stacks as follows: 

   procedure Assign (Target : in out Stack; 
                                  Source : Stacks.Stack’Class) 
   is 
      C : Stacks.Cursor’Class :=  
                                   Source.Top_Cursor;  -- dispatches 
   begin 
      Target.Clear; 
       
      for I in reverse 1 .. Source.Length loop 
         Target.Elements (I) := C.Element;  -- dispatches 
         C.Next;  -- dispatches 
      end loop; 
       
      Target.Top_Index := Source.Length; 
   end Assign; 

The Source parameter has a class-wide type, which means the 
call to Top_Cursor dispatches (since Top_Cursor is primitive 
for the type). This is exactly what we want, since different 
stack types will have different representations, and will 
therefore require different kinds of cursors. The cursor object 
(here, C) returned by the factory function is itself class-wide, 
which means that cursor operations also dispatch. The function 
call C.Element returns the element of Source at the current 
position of the cursor, and C.Next advances the cursor to the 
next position (towards the bottom of the stack). 

Ada Gem -9: Classwide Operations, 
Iterators, and Generic Algorithms 
Matthew Heaney, On2 Technologies 
Date: 17 September 2007 
 
Abstract: A generic algorithm manipulates container elements 
in a way that is completely general, working with any 
container (including arrays) that provides a way to iterate over 
its elements. In this example we systematically alter an 
operation for copying containers within a common class, 
converting it to a generic algorithm that ultimately works for 
any kind of container.  
 
Let’s get started… 
In the last gem, we used a stack class to demonstrate factory 
functions (to construct iterator objects), and implemented an 
assignment operation that dispatched on the type of the target 
stack. We mentioned in passing that that operation could be 
implemented by dispatching on the source stack, so let’s show 
how to do that. 

We reorder the parameters so that the Source stack is first in 
the parameter list (so that it’s the “distinguished receiver” of a 
prefix-style call), and change its type from classwide to 
specific. We also change the name of the operation from 
Assign to Copy, per convention. The new declaration is as 
follows: 

   procedure Copy (Source : Stack; 
                               Target : in out Stack’Class)  

   is abstract; 

In the earlier example, we had to populate the target stack in 
reverse, so that the elements would be in the correct order. We 
were able to do that because the operation was implemented 
by the specific type, and hence it had direct access to the 
representation of the (target) stack. Here the target type is 
classwide, so the only way to populate it is in forward order, 
using Push. That means we’ll have to iterate over the source 
stack in reverse, so that the items are properly ordered in the 
target. 
The bounded stack type is implemented as an array, so 
implementing Copy is easy because the bottom of the stack 
begins at the beginning of the array: 

   procedure Copy 
            (Source : Stack;  -- bounded stack (array-based) 
             Target : in out Stacks.Stack’Class) 
   is 
   begin 
      Target.Clear; 
 
      for I in 1 .. Source.Top_Index loop    -- from bottom  
                                                                  -- to top 
         Target.Push (Source.Elements (I));  -- Elements is  
                                                                   -- the array 
      end loop; 
   end Copy; 

We also said in the earlier gem that the operation need not be 
primitive for the type. If we change the source stack’s type to 
classwide, then the operation itself becomes classwide: 

   procedure Copy2  -- classwide op, not primitive 
     (Source : Stack’Class; 
      Target : in out Stack’Class); 

If we make the type of the source stack classwide, then we’ll 
need a different way to iterate over items of the source stack in 
reverse order, since we don’t have access to its representation 
anymore. 
To do that we’ll amend the cursor type to include some 
additional operations. First we’ll add a new factory function, 
to construct a cursor object that (initially) designates the 
element at the bottom of the stack: 

   function Bottom_Cursor 
                 (Container : not null access constant Stack) 
     return Cursor’Class is abstract; 

We’ll also need an operation to move the cursor to the element 
that precedes the current item: 

   procedure Previous (Position : in out Cursor)  

   is abstract; 

That gives us everything we need to turn Copy into a 
classwide operation, so that it only needs to be implemented 
once: 

   procedure Copy2 (Source : Stack’Class; 
                                 Target : in out Stack’Class) 
   is 
      C : Cursor’Class := Bottom_Cursor (Source’Access); 
   begin 
      Target.Clear; 
       
      while C.Has_Element loop 
         Target.Push (C.Element); 
         C.Previous; 
      end loop; 
   end Copy2; 

Note that we declared the classwide stack operation in the root 
package (see stacks.ads), but it could have just as easily been 
declared as a generic child procedure: 

  generic 
  procedure Stacks.Generic_Copy3 
           (Source : Stack’Class; 
            Target : in out Stack’Class); 

Actually, we could move the operation out of the package 
hierarchy entirely: 

  with Stacks; 
  generic 
     with package Stack_Types is new Stacks (<>); 
     use Stack_Types; 
  procedure Generic_Stack_Copy4 
    (Source : Stack’Class; 
     Target : in out Stack’Class); 

We can generalize this even more, such that the copy 
algorithm works for any kind of stack: 

   generic 
      type Stack_Type (<>) is limited private; 
      type Cursor_Type (<>) is private; 
      type Element_Type (<>) is private; 
 
      with function Bottom_Cursor  (Stack : Stack_Type)  
           return Cursor_Type is <>; 
      with procedure Push (Stack : in out Stack_Type; 
                                            Item  : Element_Type) is <>; 
      with procedure Previous   
            (Cursor : in out Cursor_Type)   is <>; 
     … 
  procedure Generic_Stack_Copy5  
             (Source : Stack_Type; 
              Target : in out Stack_Type); 

This illustrates the difference between the dynamic 
polymorphism of tagged types and the static polymorphism of 
generics. There is no need for a stack class anymore (having a 
dedicated copy operation that works only for types in that 
class), since the generic algorithm works for any stack. (This is 
exactly how the standard container library is designed. 
Container types are tagged, but they are not members of a 
common class.) 
Instantiating this operation on our stack type is easy, since the 
names of the generic actual operations match the names of the 

generic formal operations, so we don’t need to specify them 
explicitly (since the generic formals are marked as accepting a 
<> default): 

   procedure Test_Copy5 (S : Stack) is 
      procedure Copy5 is 
         new Generic_Stack_Copy5   
                           (Stack, 

              Cursor, 
                             Integer); -- default everything else 
 
      T : Stack (S.Length); 
   begin 
      Copy5 (Source => S, Target => T); 
   end; 

But why stop there? We can write a generic copy algorithm for 
any kind of container. We just need to generalize iteration a 
little, to mean “visit these items in the way that makes sense 
for this source container,” and generalizing insertion, to mean 
“add this element in the way that makes sense for this target 
container.” The declaration would be: 

   generic 
      type Container_Type (<>) is limited private; 
      type Cursor_Type (<>) is private; 
      type Element_Type (<>) is private; 
 
      with function First (Container : Container_Type)  
           return Cursor_Type is <>; 
      with procedure Insert  
            (Container : in out Container_Type; 
             Item      : Element_Type) is <>; 
      with procedure Advance (Cursor : in out Cursor_Type)  
       is <>; 
 
   procedure Generic_Copy6  
             (Source : Container_Type; 
              Target : in out Container_Type); 

We can instantiate this using our stack type, but note that the 
generic actuals no longer match the generic formals, so we 
need to specify them explicitly: 

   procedure Test_Copy6 (S : Stack) is 
      procedure Copy6 is 
         new Generic_Copy6 (Stack, Cursor, Integer, 
                                            First         => Bottom_Cursor, 
                                            Insert       => Push, 
                                            Advance  => Previous); 
 
      T : Stack (S.Length); 
 
   begin 
      Copy6 (Source => S, Target => T); 
   end; 

One assumption we’ve made here is that the source and target 
containers have the same type. Suppose we would like to copy 
the items in a stack to, say, an array. One approach would be 
to introduce another generic formal container type (a “source 
container” type that is distinct from the “target container” 
type), but there’s another way. Consider the implementation of 
the copy algorithm: 

   

   procedure Generic_Copy6 
            (Source : Container_Type; 
             Target : in out Container_Type) 
   is 
      C : Cursor_Type := First (Source); 
   begin 
      Clear (Target); 
      while Has_Element (C) loop 
         Insert (Target, Element (C)); 
         Advance (C); 
      end loop; 
  end Generic_Copy6; 

Notice that the only thing we do with the source container is to 
use it to construct a cursor. If we pass in the cursor directly, 
that eliminates any mention of the source stack, which in turn 
allows the source and target containers to be different types. 
Our algorithm now becomes: 

  generic 
      type Container_Type (<>) is limited private; 
      type Cursor_Type (<>) is private; 
      type Element_Type (<>) is private; 
      … 
  procedure Generic_Copy7 
           (Source : Cursor_Type; 
            Target : in out Container_Type); 

We can now copy from an integer stack to an array like this: 

   procedure Copy_From_Stack_To_Array  
           (S : in out Stack)    
   is 
      T : Integer_Array (1 .. S.Length); 
      I : Positive := T’First; 
 
      procedure Insert 
              (Container : in out Integer_Array; 
                Item      : Integer) 
      is 
      begin 
         Container (I) := Item; 
         I := I + 1; 
      end; 
 
      procedure Copy7 is 
          new Generic_Copy7   (Integer_Array,  Cursor, 
                                              Integer, 
                                              Advance => Next); 
   begin 
      Copy7 (Source => S.Top_Cursor, Target => T); 
   end Copy_From_Stack_To_Array; 

The target “container” is just an array. The only special thing 
we need to do is synthesize an insertion operation, to pass as 
the generic actual. We can also use the same algorithm to go 
the other way, from an array to a stack: 

   procedure Copy_From_Array_To_Stack  
           (S: Integer_Array)  
   is 
      T : Stack (S’Length); 
 
      function Has_Element (I : Natural) return Boolean is 

      begin 
         return I > 0; 
      end; 
 
      function Element (I : Natural) return Integer is 
      begin 
         return S (I); 
      end; 
 
      procedure Advance (I : in out Natural) is 
      begin 
         I := I - 1; 
      end; 
 
      procedure Copy7 is 
          new Generic_Copy7 (Stack, Natural,  Integer, 
                                             Insert => Push); 
 
   begin 
      Copy7 (Source => S’Last, Target => T); 
   end Copy_From_Array_To_Stack; 

Now the source container is an array, and the “cursor” is just 
the array index (an integer subtype). We have the familiar 
problem of ensuring that the target stack is populated in the 
correct order. As before, we simply iterate over the array in 
reverse, by passing the index S’Last as the initial cursor value, 
and then “advancing” the cursor by decrementing the index 
value. 
The algorithm can be generalized further still. In this final 
version, we eliminate the generic formal element type. That 
means we’ll need to modify the generic formal Insert 
operation, by passing the source cursor as a parameter instead 
of the source element. The declaration of the generic algorithm 
now becomes: 

   generic 
      type Container_Type (<>) is limited private; 
      type Cursor_Type (<>) is private; 
 
      with procedure Insert 
              (Target : in out Container_Type; 
               Source : Cursor_Type) is <>; 

 
      … 
   procedure Generic_Copy8 
            (Source : Cursor_Type; 
             Target : in out Container_Type); 

The algorithm is now agnostic about the mapping from cursor 
to element (since it doesn’t even know about elements), which 
is more flexible, since it allows the client to choose whatever 
mechanism is the most efficient. To use the new algorithm, all 
we need to do is make a slight change to the generic actual 
Insert procedure, as follows: 

   procedure Copy_From_Stack_To_Array  
          (S : in out Stack)  
   is 
      T : Integer_Array (1 .. S.Length); 
      I : Positive := T’First; 
 
      procedure Insert 
              (Target : in out Integer_Array; 
               Source : Cursor)  -- now a cursor instead of  
                                            -- an element 
      is 
      begin 
         Target (I) := Element (Source); 
         I := I + 1; 
      end; 
 
      procedure Copy8 is 
         new Generic_Copy8 (Integer_Array, Cursor, 
                                           Advance => Next); 
   begin 
      Copy8 (Source => S.Top_Cursor, Target => T); 
   end Copy_From_Stack_To_Array; 

The basic idea is that a generic algorithm can be used over a 
wide range of containers (including array types). A cursor 
provides access to the elements in a container, but as we’ve 
seen, once you have a cursor then the container itself sort of 
disappears. From the point of view of a generic algorithm, a 
container is merely a sequence of items. 

